

' MAIN FORM CODE

```
Option Explicit

Const MAXCITIES = 20
Const MAXMONTHS = 12

Dim City(1 To MAXCITIES) As String
Dim Rain(1 To MAXCITIES, 1 To MAXMONTHS) As Integer
Dim CityAvr(1 To MAXCITIES) As Integer
Dim MonthAvr(1 To MAXMONTHS) As Integer
Dim NumCity As Integer

Private Sub mnuCalculate_Click()
    Dim LowCity As Integer
    Dim LowMonth As Integer

    LowCity = 1
    LowMonth = 1

    FindResults LowCity, LowMonth, Rain(), NumCity, CityAvr(), MonthAvr(), MAXMONTHS
    DisplayData picRainList, City(), Rain(), CityAvr(), MonthAvr(), NumCity, MAXMONTHS
    DisplayLowRain picRainList, LowMonth, LowCity, City(), Rain()
End Sub

Private Sub mnuOpen_Click()
    Dim FileName As String

    NumCity = 0

    FileName = GetFile(cdlDialog)
    If FileName <> "" Then
        NumCity = ReadData(FileName, City(), Rain(), MAXMONTHS)
        DisplayData picRainList, City(), Rain(), CityAvr(), MonthAvr(), NumCity, MAXMONTHS
        mnuCalculate.Enabled = True
    End If
End Sub
```

' CODE MODULE CODE

```
Option Explicit

Public Function ReadData(ByVal FName As String, CityName() As String, Rainfall() As Integer, ByVal MAXMONTHS As Integer) As Integer
    Dim X As Integer
    Dim K As Integer

    K = 0
    Open FName For Input As #1
    Do While Not EOF(1)
        K = K + 1
        Input #1, CityName(K)
        For X = 1 To MAXMONTHS
            Input #1, Rainfall(K, X)
        Next X
    Loop
    Close #1
    ReadData = K
End Function

Public Sub DisplayData(PicBox As PictureBox, CityName() As String, Rainfall() As Integer, CityAvg() As Integer, MonthAvg() As Integer, _
                                                                    ByVal Num As Integer, ByVal MAXMONTHS As Integer)
    Dim X As Integer
    Dim Y As Integer
    Dim Msg As String

    PicBox.Cls
    PicBox.Print "City        Jan Feb Mar Apr May Jun Jul Aug Sep Oct Nov Dec   Avg"
    PicBox.Print
    For X = 1 To Num
        Msg = Format$(CityName(X), "!@@@@@@@@@@@")
        For Y = 1 To MAXMONTHS
            Msg = Msg & _
                Format$(Rainfall(X, Y), "@@@@")
        Next Y
        Msg = Msg & Format$(CityAvg(X), "@@@@@@")
        PicBox.Print Msg
    Next X
    PicBox.Print
    PicBox.Print "Monthly Avr";
    For Y = 1 To MAXMONTHS
        PicBox.Print Format$(MonthAvg(Y), "@@@@");
    Next Y
    PicBox.Print
End Sub
```

```
Public Sub FindResults(CityLow As Integer, MonthLow As Integer, Rainfall() As Integer, ByVal Num As Integer, CityAvg() As Integer, _
                                                                                    MonthAvg() As Integer, ByVal MAXMONTHS As Integer)
    Dim X As Integer
    Dim Months As Integer
    Dim Lowest As Integer
    Dim Monthly As Integer
    Dim CitySum As Integer

    Lowest = Rainfall(1, 1)

    For X = 1 To Num                                ' Looping through each city
        CitySum = 0
        For Months = 1 To MAXMONTHS                 ' Looping through each month
            CitySum = CitySum + Rainfall(X, Months)
            If Rainfall(X, Months) < Lowest Then    ' Checking for lowest rain amount
                Lowest = Rainfall(X, Months)        ' Amount of rain
                CityLow = X                         ' City number
                MonthLow = Months                   ' Month number of lowest amount
            End If
        Next Months
        CityAvg(X) = CitySum / MAXMONTHS            ' Calculating average for city
    Next X

    For Months = 1 To MAXMONTHS                     ' Calculating overall monthly averages
        Monthly = 0
        For X = 1 To Num
            Monthly = Monthly + Rainfall(X, Months)
        Next X
        MonthAvg(Months) = Monthly / Num
    Next Months
End Sub

Public Function GetFile(Dialog As CommonDialog) As String
    Dialog.FileName = ""
    Dialog.Filter = "Text Files|*.txt"
    Dialog.InitDir = App.Path
    Dialog.ShowOpen
    GetFile = Dialog.FileName
End Function

Public Sub DisplayLowRain(PicBox As PictureBox, ByVal MonthLow As Integer, ByVal CityLow As Integer, CityName() As String, _
                                                                                                Rainfall() As Integer)
    Dim Month_Name As String

    Month_Name = Format$(CDate(Str$(MonthLow) & Str$(MonthLow) & "/00"), "mmmm") ' LiangChao method
    PicBox.Print
    PicBox.Print "Least rain fell in " & CityName(CityLow) & " during " & Month_Name & " (" & _
                                                        LTrim$(Str$(Rainfall(CityLow, MonthLow))) & " units)"
End Sub
```